Shooting 法による Falkner-Skan 方程式の解法 with Scilab

伊藤榮信

2017 年 1 月 11 日

1　Shooting 法と Scilab プログラム

より一般的な境界層方程式である Falkner-Skan 方程式

$$f''' + \beta_0 f f'' + \beta(1 - f'^2) = 0, \tag{1}$$

を数値的に解くために Shooting 法を利用する（以下，$\beta_0 = 1$ とする）．ここで $f'(\eta) = \mathrm{d}f(\eta)/\mathrm{d}\eta$ を表し，2 階，3 階も同様である．境界条件は

$$f(0) = 0, \ \ f'(0) = 0, \qquad \eta \to \infty \ \ \text{で} \ \ f' \to 1 \tag{2}$$

である．まず Falkner-Skan 方程式を 1 階連立常微分方程式

$$\left.\begin{aligned} \frac{\mathrm{d}f}{\mathrm{d}\eta} &= u, \\ \frac{\mathrm{d}u}{\mathrm{d}\eta} &= v, \\ \frac{\mathrm{d}v}{\mathrm{d}\eta} &= -fv - \beta(1 - u^2), \end{aligned}\right\} \tag{3}$$

の形にしておく．よって境界条件は

$$f(0) = 0, \quad u(0) = 0, \quad u(\infty) = 1 \tag{4}$$

となる．

境界条件の一つは $\eta \to \infty$ でのものであるが，未知の s を導入し，すべて $\eta = 0$ での条件として解法を試みる：すなわち

$$f(0) = 0, \quad u(0) = 0, \quad v(0) = s \tag{5}$$

として解きはじめ（例えば常微分方程式の解法には Runge-Kutta を使う）$\eta \to \infty$ で $u \to 1$ となるような，s を探索するという問題に帰着する．つ

まり，s を修正しながら $\eta \to \infty$ で $u \to 1$ なる s を探す．最初の s を s_0 として第 n 回目を s_n と書く．そこで問題となるのが s_n の更新方法

$$s_{n+1} = s_n + \Delta s \tag{6}$$

であり，それではどのように Δs を決めるかとなる．もし s_{n+1} で $u(s_{n+1}) = 1$ が達成されたとすると，1 次精度で

$$u(s_{n+1}) = u(s_n) + \frac{\partial u}{\partial s}\Delta s = 1 \tag{7}$$

という式が得られる．Δs はこの式から求められ

$$\Delta s = (1 - u(s_n))\frac{\partial s}{\partial u}. \tag{8}$$

となる．ここで

$$\frac{\partial s}{\partial u} = \frac{s_n - s_{n-1}}{u(s_n) - u(s_{n-1})} \tag{9}$$

とする．よって，

$$s_{n+1} = s_n + \Delta s = s_n + (1 - u(s_n))\frac{s_n - s_{n-1}}{u(s_n) - u(s_{n-1})} \tag{10}$$

で更新し，s_n が収束するまで繰り返すことで解を見つけるのが shooting 法である．

以下，Scilab によるプログラムを記しておく．Scilab での常微分方程式解法機能は “ode” である．

```
//////////////Falkner-Skan solver ///////////////////////////
// ODE :     f”’+ff”+beta*(1-f’2)=0   → beta=bb
// written by Itoh Shigenobu
clear;
format(20) //17 decimal places
bb=1;
// eta=x, f=y(1), u=y(2), v=y(3)
function ydot=FLK_SK(x,y)
   ydot(1)=y(2);
   ydot(2)=y(3);
   ydot(3)=-y(1)*y(3)-bb*(1-y(2)*y(2));
endfunction

x_start=0;
x_step=0.2;
x_end=4.0;
```

```
n_end=(x_end-x_start)/x_step+1;
epsilon=0.00001;// control parameter for convergence

// n=1 making initial state
shooting_value(1)=1; //initial shooting parameter s_0
x=x_start:x_step:x_end; // interval[0, 4]
yy=ode([0;0;shooting_value(1)],x_start,x,FLK_SK);
f_prime_at_end(1)=yy(2,n_end)
shooting_value(2)=1.1; // shooting parameter for n=2

n=2
x=x_start:x_step:x_end; // interval[0, 3]
while abs(shooting_value(n)-shooting_value(n-1)) > epsilon
   yy=ode([0;0;shooting_value(n)],x_start,x,FLK_SK);
   f_prime_at_end(n)=yy(2,n_end);
   shooting_value(n+1)=shooting_value(n)+(1-f_prime_at_end(n))*(shooting_value(n)-
shooting_value(n-1))/(f_prime_at_end(n)-f_prime_at_end(n-1));
n=n+1;
end

/////////////////// Graph
for i=1:1:n_end,
   X(i)=x_start+(i-1)*x_step
end

for i=1:1:n_end,
   Y2(i)=yy(2,i) // yy(2,i)=u(i)
end

for i=1:1:n_end,
   Y3(i)=yy(3,i) // yy(3,i)=v(i)
end

clf();
plot(Y2,X,"o")
//////////////Falkner-Skan solver end///////////////////
```

2　いくつかの数値結果

Falkner-Skan 方程式内の β のいくつかの値に対する結果を記しておく．文献では各 β に対する $v(0) = s$ の値により精度を確認している．これまでの数値実験によれば，この値は無限遠としてとる有限値 η_∞（プログラム内では x_end）の与え方により若干増減する．本プログラムでは $\beta > 1.8$ では解が得られなかった．また，η の刻み幅（x_step）は 0.2 とかなり粗いように思われるが，0.01 でも 0.0001 でも $v(0)$ の値はほとんど変化しない．以上の結果と他の文献 Asaithambi[1] との比較を表 1 に示した．また $\beta = 1.0, 0, -0.1988$ の 3 ケースの速度プロフィール（$f' = u$ の結果，縦軸が η で横軸が $f' = u$ の値）を載せた．$\beta = -0.1988$ の場合は速度プロフィールに変曲点を持つ流れである．

表 1: 数値実験結果と他の文献との比較

β	今回 η_∞	今回 $f'''(0) = v(0)$	Asaithambi(2005)
2.0	-	-	1.687218
1.8	2.6	1.606617	-
1.0	4.0	1.232595	1.232589
0.5	4.6	0.927687	0.927680
0.0	6.0	0.469599	0.469600
-0.1	8.0	0.319270	0.319270
-0.15	8.0	0.216361	0.216361
-0.18	8.0	0.128636	0.128637
-0.1988	8.0	0.005232	0.005225



図 1: $\beta = 1.0$ の場合の $f' = u$(速度プロフィール)



図 2: $\beta = 0.0$ の場合の $f' = u$(速度プロフィール)



図 3: $\beta = -0.1988$ の場合の $f' = u$(速度プロフィール)

参考文献

[1] Asaithambi, A.: Solution of the Falkner-Skan equation by recursive evaluation of Taylor coefficients, Journal of Computational and Applied Mathematics 176(2005): pp203-214.